

0.9 ジュニア版 乱数とその利用

これまでも少しでてきましたが、コンピュータ利用と乱数は密接な関係があります。ここでは、乱数の基本となる「一様分布」「標準正規分布」「ガンマ分布」の乱数を取り上げて、特に一般的な一様分布及び正規分布への拡張とその実地的な使い方を解説します。

GAUSS における乱数



プログラム


```
new;  
cls;  
x=rndu(10000,1);  
y=rndn(10000,1);  
a=2;  
z=rndgam(10000,1,a);  
library pgraph;  
graphset;  
pqgwin many;  
call hist(x,19);  
call hist(y,19);  
call hist(z,19);
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 組込み関数 `rndu` で 0 と 1 の間の一様乱数を 10000×1 (すなわち 10000 個) だけ発生させて、その結果を変数 `x` に代入する。
- 4 行目 組込み関数 `rndu` で 0 と 1 の間の一様乱数を 10000×1 (すなわち 10000 個) だけ発生させて、その結果を変数 `y` に代入する。
- 5 行目 変数 `a` に 2 を代入する。
- 6 行目 組込み関数 `rndgam` でパラメータ `a` のガンマ分布にしたがう乱数を 10000×1 (すなわち 10000 個) だけ発生させて、その結果を変数 `z` に代入する。
- 7 行目 グラフを描く `pgraph` のライブラリを呼び出す。
- 8 行目 その `pgraph` のライブラリのグローバル変数 (初期設定) を初期化する。
- 9 行目 グラフの描かれる PQG ウィンドウが一度に複数開けるように設定を変更する。
(この設定がなければ、GAUSS のデフォルトは一度に 1 枚のグラフのみを表示)
- 10 ~ 12 行目 `x` (`y` および `z`) のヒストグラムを 19 本の棒グラフで描く。

グラフ表示





上のグラフの作り方と同じものについて、それぞれの平均と標準偏差を表示するプログラムは次のようになる（ただし、乱数に依存しているなのでその値はその時々で異なる）。

プログラム

```
new;
cls;
x=rndu(10000,1);
y=rndn(10000,1);
a=2;
z=rndgam(10000,1,a);
print "    U[0,1)" meanc(x) stdc(x);
print "    N(0,1)" meanc(y) stdc(y);
print "Gamma(a=2)" meanc(z) stdc(z);
```

プログラム各行の説明

1～6行目 （ 同上 ）

7～9行目 引用符内のメッセージの文字列とともに、 x （ y および z ）の平均と標準偏差を計算結果の2つを画面表示する。

画面表示

U[0,1)	0.50191778	0.28951933
N(0,1)	0.0053559613	1.0028955
Gamma(a=2)	1.9946959	1.4112971

一様乱数の引き伸ばしと平行移動


プログラム

```
new;  
cls;  
x=6*randu(10000,1);  
library pgraph;  
graphset;  
call hist(x,19);
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 組込み関数 `randu` で 0 と 1 の間の一様乱数を 10000×1 （すなわち 10000 個）だけ発生させて、それに 6 倍したその結果（すなわち、0 と 6 の間の一様乱数）を変数 `x` に代入する。
- 4 行目 グラフを描く `pgraph` のライブラリを呼び出す。
- 5 行目 その `pgraph` のライブラリのグローバル変数（初期設定）を初期化する。
- 6 行目 `x` のヒストグラムを 19 本の棒グラフで描く。

グラフ表示



上のように、これまでは一様乱数は 0 と 1 の間であったが、それが 6 倍されてその分一様に引き伸ばされて 0 と 6 の間の一様乱数になっていることがわかる。

今度は、一様乱数の引き伸ばしに加えて平行移動させてみる。


プログラム

```
new;  
cls;  
x=2*randu(10000,1)-1;  
library pgraph;  
graphset;  
call hist(x,19);
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 組込み関数 `randu` で 0 と 1 の間の一様乱数を 10000×1 （すなわち 10000 個）だけ発生させて、それに 2 倍したその結果（すなわち、0 と 2 の間の一様乱数）から 1 を引いたもの（すなわち、- 1 と 1 の間の一様乱数）を変数 `x` に代入する。
- 4 行目 グラフを描く `pgraph` のライブラリを呼び出す。
- 5 行目 その `pgraph` のライブラリのグローバル変数（初期設定）を初期化する。
- 6 行目 `x` のヒストグラムを 19 本の棒グラフで描く。

グラフ表示



標準正規乱数の引き伸ばしと平行移動（一般的な正規乱数の作成）

記述統計量の章でデータの標準化を少し扱いました。データの標準化とは、データ X を

$$Z = \frac{X - \mu}{\sigma}$$

というふうに、平均 μ を引くことで平均を 0 に中央化して、標準偏差 σ で割ることによって新しく作成されたデータの散らばりを表す標準偏差を 1 に標準化するものでした。したがって、どのような分布のデータも上の μ を引いて σ で割れば、平均 0 標準偏差 1 になります。これを逆方向に利用しましょう。例えば、 $N(0,1)$ の標準正規分布から始まって、その乱数を作成する関数自体にある数をかけたとしましょう。例えば、下の例では 5 をかけます。そうするとデータの標準化では 5 で割りましたが、ここではすでに標準正規分布に 5 をかけます。そうすると、標準化の類推からも容易に見当がつくように標準偏差が 5 に引き伸ばされた分布ができるはずです。

プログラム

```
new;  
cls;  
x=5*randn(10000,1);  
print "N(0,25)" meanc(x) stdc(x);  
library pgraph;  
graphset;  
call hist(x,19);
```


プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 今度は組込み関数 `randn` で平均 0 と分散 1 の標準正規乱数を 10000×1 （すなわち 10000 個）だけ発生させて、それに 5 倍したその結果（すなわち、平均 0 と標準偏差 5（分散 5^2 ）の正規乱数）を変数 x に代入する。
- 4 行目 引用符の中のメッセージとともに、 x の平均と標準偏差を計算した結果を 2 つ並べて画面表示する。
- 5 行目 グラフを描く `pgraph` のライブラリを呼び出す。
- 6 行目 その `pgraph` のライブラリのグローバル変数（初期設定）を初期化する。
- 7 行目 x のヒストグラムを 19 本の棒グラフで描く。

画面表示

N(0,25) 0.019377683 4.9931309

グラフ表示



同様に、今度は上の操作に加えて、平均を平行移動することによって変更してみましょう。標準偏差は5に伸ばしたままで、平均を0から10に平行移動させます。ちょうど、平均が10のデータを10を引くことによって標準化する場合の逆方向を考えるわけです。すなわち、平均を10にするにはただ単に引き伸ばした分布にその平均の数10を足すだけです。括弧をつけて5で引き伸ばす前の乱数を平行移動させてはいけません。なぜならば、万が一そうすると引き伸ばす標準偏差を最後にかけることになってしまって、平行移動の数が10ではなくなってしまうからです。したがって、ただ単にその数を足して平行移動するだけで足ります。

プログラム

```
new;  
cls;  
x=5*randn(10000,1)+10;  
print "N(10,25)" meanc(x) stdc(x);  
library pgraph;  
graphset;  
call hist(x,19);
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 今度は組込み関数 `randn` で平均 0 と分散 1 の標準正規乱数を 10000×1 （すなわち 10000 個）だけ発生させて、それに 5 倍して 10 加えた結果（すなわち、平均 10 と標準偏差 5（分散 5^2 ）の正規乱数）を変数 `x` に代入する。

4 行目 引用符の中のメッセージとともに、x の平均と標準偏差を計算した結果を 2 つ並べて画面表示する。

5 行目 グラフを描く pgraph のライブラリを呼び出す。


6 行目 その pgraph のライブラリのグローバル変数（初期設定）を初期化する。

7 行目 x のヒストグラムを 19 本の棒グラフで描く。

画面表示

N(0,25) 10.088126 5.0189733

グラフ表示



上のように、ヒストグラムの中心は 0 から 10 に移動していて、乱数の平均がほぼ 10 になっていることがわかります。

すなわち、標準偏差 をかけて平均を加える

$$N(0,1) \rightarrow \sigma = 5, \mu = 10 \rightarrow \sigma N(0,1) + \mu$$

という操作を標準正規乱数 $N(0,1)$ に加えていることになります。

一般的な一様乱数および正規乱数の関数化

プログラムのなかでこれまでのような操作を直接してもかまいませんが、見通しをよくするために 1 行定義関数 fn の機能を用いてそれぞれの一般形を関数として作成しておきましょう。正規乱数の場合は簡単で、上の変換関係をそのまま関数で表します。すなわち、関数に適当に名前をつけて) ここでは rndnorm としてやろう)

```
fn rndnorm(r,c,m,s)=s*rndn(r,c)+m;
```

というふうに、組込み関数 rndn のインプットの r と c (乱数の行数と列数) とともに平均

mと標準偏差 s の合計 4 つをインプットにしてやって、もともとの `rndn(r,c)` に標準偏差 s をかけて平均 m を加えたものを関数名にイコールサインで定義しています。ここで m がマイナスの値をとるとその値がそのまま新しく作成された正規乱数の平均になります。なお標準偏差は分布のちらばりを表す尺度であって、マイナスの値をとることはできません。一般化した一様乱数は少し面倒です。このようなときにこそ、関数を作っておけば後で簡単に利用できます。ここで関数名を適当に定めて（ここでは `rndunif` としてやろう）その新しい一様乱数の下限を `low`、上限を `up` としてやりましょう。少し前に行なった一様乱数の変換は、0 と 1 の間の一様乱数を 2 倍して 1 を引いたもの（- 1 を加えたもの）が、結果として - 1 と 1 の間の一様乱数になるというものでした。同様に考えて、もとの 0 と 1 の間の一様乱数を 3 倍して 1 を加えると、0 と 3 の間の一様乱数を 1 平行移動したもの、すなわち、1 と 4 の間の一様乱数になるはずですが、これらを新しく作られる一様乱数の下限 `low` と上限 `up` に注目してあらためて考えなおしてやると、1 と 4 の距離（すなわち `up-low`）倍したものを下限 `low` だけ平行移動したものになるはずですが、すなわち、`rndu` のもともとのインプット `r` と `c`（行数と列数）とともに下限 `low` と上限 `up` の 4 つをインプットとして

```
fn rndunif(r,c,low,up)=(up-low)*rndu(r,c) +low;
```

この関係を `rndu(r,c)` に対して `up` と `low` の距離 (`up-low`) をかけて、下限 `low` を加えるという `(up-low)*rndu(r,c) +low` という式を関数名にイコールサインでつないで定義しています。実際、最初の例では、下限は - 1 上限は 1 でその距離は 2 で、下限の - 1 だけ平行移動したものと等しくなっていて、この考え方は合っていることがわかります。

プログラム

```
new;
cls;
x=rndunif(10000,1,5,10);
y=rndnorm(10000,1,10,5);
print " U[5,10]" meanc(x) stdc(x) minc(x) maxc(x);
print "N(10,25)" meanc(y) stdc(y);
```

```
fn rndunif(r,c,low,up)=(up-low)*rndu(r,c) +low;
```

```
fn rndnorm(r,c,m,s)=s*rndn(r,c)+m;
```

画面表示

U[5,10)	7.5129421	1.4416948	5.0001635	9.9997065
N(10,25)	9.9674280	5.0209938		

一様乱数の応用（判断分岐）

ふたたび 0 と 1 の間の一様乱数に戻って、これを用いて二者択一の判断分岐をしよう。すなわち、一様乱数で出てきた数が 0.5 よりも小さければ第 1 の選択をし、そうでなければ（0.5 以上であれば）第 2 の選択をするというものである。この章の一番最初に示した 0 と 1 の間の一様乱数のヒストグラムを見ても明らかなように、ここでは、第 1 の選択と第 2 の選択の起こる確立は回数を重ねてならしてみれば、ほぼ二分の一ずつになるはずであることを利用している。

プログラム

```
new;  
cls;  
x=randu(1,1);  
if x<0.5;  
    print "Decision 1";  
else;  
    print "Decision 2";  
endif;
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 一様乱数を 1 × 1 の行列分（すなわち 1 個）作成して変数 x に入れる
- 4 行目 x が 0.5 よりも小さければ、次の命令を実行する
- 5 行目 Decision 1 と画面表示する
- 6 行目 そうでなければ（x が 0.5 以上であれば）次の命令を実行する
- 7 行目 Decision 2 と画面表示する
- 8 行目 I f 条件分岐の最後の部分（I f を使えば必ず必要）

一様乱数の応用（整数乱数の作成）

サイコロを振ると 1 から 6 までのなかの整数のどれかの目ができる。しかも回数を重ねてならしてみるとそれぞれの目の出る確率は 6 分の 1 になるはずである。このことを一様乱数でやってみよう。やり方は、0 と 1 の間の一様乱数を 6 倍することによって、0 と 6 の間の一様乱数にしておいて、最後に小数点以下を切り上げてやれば、{1,2,3,4,5,6}のなかのどれかの整数になるはずである。

プログラム

```
new;  
cls;  
x=6*randu(10,1);
```

```
print ceil(x);
```

画面表示

```
3.0000000  
5.0000000  
5.0000000  
3.0000000  
2.0000000  
2.0000000  
3.0000000  
1.0000000  
4.0000000  
4.0000000
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 10×1 のディメンション（すなわち 10 個の要素からなる列ベクトル）の 0 と 1 の間の一様乱数を 6 倍したもの（0 と 6 の間の一様乱数のこと）を x に入れる
- 4 行目 x を切り上げの組込み関数 `ceil` によって変換されて出てきたものを画面表示する

単純であるので、

```
print ceil(6*randu(10,1));
```

などと 1 行で示してやってもよいだろう。1 から 6 までの整数ではなくて、1 から n までの整数がほしい場合には 6 のところを n に変更するだけでよい。なお、`ceil` は英語で天井を意味する。同様に、GAUSS では切捨ては `floor`、四捨五入は `round` をそれぞれ使う。

一様乱数の応用（データの中から 1 つピックアップする）

実際のデータの要素が n 個あって、その中から 1 つランダムにピックアップするにはどうしたらよいだろうか？それは、1 から n までの整数の乱数を作成して、実際のデータのその番目（行番号）のデータを示せばよい。

プログラム

```
new;  
cls;  
data={78,68,89,92,60,42,74,80,78};  
n=rows(data);
```

```
index=ceil(n*randu(1,1));  
print data[index];
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 変数 **data** に 9 行 1 列の{78,68,89,92,60,42,74,80,78}を入れる
- 4 行目 **data** の行数を調べて、それを変数 **n** に入れる
- 5 行目 (1 × 1 のディメンションの) 1 つの 0 と 1 の間の一様乱数に **n** をかけることで 1 から **n** までの一様乱数にしておいて、それをさらに組込み関数 **ceil** で切り上げることによって、1 から **n** までの整数の乱数にして、その結果を **index** に入れる
- 6 行目 変数 **data** の **index** 番目の数を表示する

上の例では乱数が実際のデータと結びついてデータをランダムにピックアップするという作業ができることを示しています。乱数はシミュレーションためだけのものではありません。

一様乱数の応用 (ウェイトのミックス)

今、点 $x_1(1,2)$ と点 $x_2(3,4)$ の間の間を $1 - a$ 対 a に分割する点を

$$(1 - a) x_1 + a x_2$$

と考えると、 a を適当に 0 から 1 までの数で (すなわち一様乱数で) ランダムにミックスしたものにしてみましょう。同様に、 y についても同じ a を用いてランダムにミックスします。乱数 a を 1000 個発生させて、100 通りの軌跡を $x-y$ 平面上にプロットしてみます。

プログラム


```
new;  
cls;  
x1=1; y1=2;  
x2=3; y2=4;  
a=randu(100,1);  
x=a.*x1+(1-a).*x2;  
y=a.*y1+(1-a).*y2;  
library pgraph;  
graphset;  
xtics(0,5,1,0);  
ytics(0,5,1,0);
```

`xy(x,y);`

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 x_1 に 1 を代入する。 y_1 に 2 を代入する。
- 4 行目 x_2 に 3 を代入する。 y_2 に 4 を代入する。
- 5 行目 組込み関数 `randu` で 0 と 1 の間の一様乱数を 100×1 (すなわち 100 個) だけ発生させて、その結果を変数 a に代入する。ここではウエイト a をランダムにする。
- 6 行目 ウエイト a を用いて、 $a x_1 + (a - 1) x_2$ を変数 x に代入する (すなわち、 x とおく)。掛け算にドットがついているのは要素ごとの計算の意味。
- 7 行目 ウエイト a を用いて、 $a y_1 + (a - 1) y_2$ を変数 y に代入する (すなわち、 y とおく)。掛け算にドットがついているのは要素ごとの計算の意味。
- 8 行目 グラフを描く `pgraph` のライブラリを呼び出す。
- 9 行目 その `pgraph` のライブラリのグローバル変数 (初期設定) を初期化する。
- 10 行目 グラフの x 軸は最小値 0 最大値 5 で 1 ずつの大目盛、小目盛は 0 で無し。
- 11 行目 グラフの y 軸は最小値 0 最大値 5 で 1 ずつの大目盛、小目盛は 0 で無し
- 12 行目 x と y を $x-y$ グラフでプロットする。

グラフ表示



乱数のシードについて

プログラム

```
new;  
cls;  
rndseed 1000;  
print rndu(1,1);  
print rndn(1,1);
```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 行目 乱数のシードを 1000 に強制的に変更して同じ結果が出るようにする
- 4 行目 組み関数 `rndu` で 0 と 1 の間の一様乱数を 1×1 (すなわち 1 個) だけ発生させて、その結果を画面表示する
- 5 行目 組み関数 `rndn` で標準正規乱数を 1×1 (すなわち 1 個) だけ発生させて、その結果を画面表示する

GAUSS ではその立ち上げ時刻に応じて乱数を発生させるおおもとのシードという数を内部で自動的に決定しています。同じマシンでは同じ乱数結果が出ることはまずありませんが、異なるマシンではタイミングによって同じ乱数結果もでることもあります。ただし、同じ状況を作り出すには、この乱数のおおもとの数を強制的に変更してだれでもいつでもどのマシンでも一定の乱数を出す必要があります。そのために設定するのがシードとよばれるものです。乱数の組み関数を呼び出すよりも前に、

`rndseed 整数;`

というふうに書いて設定をします。この他に、乱数のおおもとなる係数部分と定数部分も設定を変えることができますが、学术论文レベルでも、通常の使用でも、まずそれらの変更はしません。このシードの変更だけをするようになります。

注意事項

- 1) シードの設定は冒頭で 1 回限りで十分である。多くとも、各乱数につき 1 つずつのシード設定をするのみで足りる。しかし、その場合もまとめて 1 回でも足りる。
- 2) 旧式の GAUSS では `rndus` と `rndns` というふうに `s` の字がついてその第 3 パラメータにシードを設定できるものもあったが、現在は廃止された。
- 3) シードを 1 から順に増していくなどといった方法では、ステートの変化を表すことはできないばかりか、そうしてしまうと理論上は乱数とは言いがたい。

練習問題

【問 1】

1 万個からなる -2 と 2 と間に一様に分布する乱数を作成して、それをグラフにしよう。

【問 2】

1 万個からなる平均 -2 「分散」 2 の正規分布にしたがう乱数を作成して、それが実際に平均 -2 分散 2 に近い値になっているかを確認した上、それをグラフにしよう。

【問 3】

標準の GAUSS の組込み関数として、他にどんな乱数があるのかを調べてみよう。

発展

Luenberger サイコロの例： [6.2a](#) [6.2b](#) [6.2c](#)
回転盤の例： [6.3a](#) [6.3b](#)